

## 9/2.3

# Le debugger

---

### Qu'est-ce qu'un debugger ?

Ce mot vient du mot anglais « bug » qui signifie erreur en français. Un debugger est un outil qui permet de faciliter la mise au point des programmes écrits en assembleur. C'est un complément indispensable aux programmes précédents : désassembleur et assembleur.

### Analyse du problème

Nous allons créer un debugger écrit dans le langage assembleur Z80 le moins complexe mais le plus complet possible. Les fonctions minimum d'un tel programme sont :

- Lecture d'un programme source (en code exécutable) par exemple créé par le programme assembleur décrit en 9/2.2 ;
- Sauvegarde sur disquette d'un programme source éventuellement modifié, ou d'une zone mémoire ;
- Exécution d'un programme ou d'une partie de programme à partir d'une adresse déterminée et jusqu'à une adresse déterminée puis affichage des registres lorsque l'adresse finale est atteinte ;
- Affichage et éventuellement modification d'une case mémoire ;
- Affichage d'une zone mémoire.

Les fonctions que nous venons d'énumérer sont accessibles par un menu affiché en début de programme et après chaque commande. Une fonction est activée en tapant la lettre du menu correspondante (par exemple L pour Lecture, X pour exécution, etc.). Passez en mode CAPS LOCK pour que le debugger fonctionne correctement. Si vous tapez une option que le debugger ne comprend pas, le menu est réaffiché et aucune autre action n'est effectuée.


Les fonctions du debugger sont développées dans le programme assembleur qui suit. Nous allons les examiner en détail.

## I. Lecture d'un programme objet

Les actions réalisées par cette option sont les suivantes :

- saisie du nom du programme à lire :

Pour cela, nous avons créé une routine de lecture (SAINOM) qui permet de lire une chaîne alphanumérique quelconque de longueur maximum 12 caractères qui se termine par le caractère CR (Carriage Return). La chaîne est stockée dans un buffer de lecture (BUFF). La routine de lecture suit la logique suivante :



- ouverture du fichier :


L'ouverture du fichier se fait en utilisant la macro du Firmware qui se trouve à l'adresse #BC77 et qui a pour nom CAS IN OPEN. A noter que cette macro est utilisable sur les trois CPC (464, 664 et 6128).

- lecture de l'adresse d'implantation :

Cette lecture est précédée de l'affichage du message "A implanter à l'adresse : \_\_".

La lecture de l'adresse d'implantation est faite en utilisant une autre routine que SAINOM pour rendre automatique le retour chariot en fin de saisie. En effet, le nombre de caractères saisis est invariablement quatre. Le

code ASCII de chaque caractère est stocké séquentiellement dans la zone BUFF. La logique de la routine SP1 créée à cet effet est la suivante :



— Lecture du fichier ouvert en lecture :

La lecture du fichier se fait en utilisant la macro du Firmware située à l'adresse #BC83 qui a pour nom CAS IN DIRECT. A noter que cette routine est identique sur les trois systèmes CPC. Sur les modèles 664 et 6128, la lecture se fait cependant sur disquette.

— Fermeture du fichier ouvert en lecture :

La fermeture du fichier se fait en utilisant la macro du Firmware située à l'adresse #BC7A qui a pour nom CAS IN CLOSE. A noter que cette routine est identique sur les trois systèmes CPC.

Comme dans toutes les options du menu, les registres AF, BC, DE et HL sont mis en pile au début de la routine et dépilés en fin de routine.


## II. Sauvegarde d'un programme objet ou d'une zone mémoire

- Les registres AF, BC, DE et HL sont sauvegardés en début de routine ;
- Le nom du fichier cassette ou disquette est demandé en utilisant la même routine (SAINOM) que pour l'option de lecture précédemment décrite ;
- Le fichier est ouvert en écriture en utilisant la macro du Firmware située en #BC8C (CAS OUT OPEN) ;
- Les adresses de début et de fin de sauvegarde sont demandées en utilisant le sous-programme de saisie SP1 ;
- Les données correspondantes sont alors envoyées vers le lecteur de cassettes ou de disquettes en utilisant la macro du Firmware située en #BC98 (CAS OUT DIRECT) ;
- Le fichier ouvert en écriture est enfin fermé en utilisant la macro du Firmware située en #BC8F (CAS OUT CLOSE).
- Les registres AF, BC, DE et HL sont dépilés et le sous-programme rend la main au menu.

## III. Exécution d'un programme

- Les registres AF, BC, DE et HL sont sauvegardés en début de routine ;
- Les adresses de début et de fin d'exécution sont lues en utilisant la routine SP1. Ces adresses sont sauvegardées dans les variables ADDEB et ADFIN ;
- Les trois octets de code commençant à l'adresse de fin d'exécution sont sauvegardés dans une zone buffer (BSTOP) ;
- Le code JP 83D0H (#C3, #D0, #83) est implanté dans cette zone. Ce code permet d'activer la routine d'affichage du contenu des registres. Voyons en détail son fonctionnement :
  - Une routine de nom ALPHA affiche toutes les constantes texte (nom des registres) sur l'écran ;
  - Une routine de nom AVAL affiche le contenu des registres. Pour cela, une routine d'affichage spécifique a été développée. Elle a pour nom HEAS. Le quartet supérieur est séparé du quartet inférieur, converti en ASCII, et affiché par une routine Firmware.

L'organigramme suivant décrit l'enchaînement des opérations de conversion et d'affichage :



— La routine d'affichage des registres rend la main à la routine EXEC qui restitue le code sauvegardé dans BSTOP, dépile les registres et rend la main au menu.

#### IV. Affichage du contenu d'une mémoire

- Les registres AF, BC, DE et HL sont sauvegardés en début de routine ;
- L'adresse de la mémoire à afficher est saisie grâce à la routine SP1 décrite précédemment ;
- Le contenu de cette mémoire est alors converti en ASCII et affiché par la routine HEAS ;
- La routine SAISIE est alors activée. Elle permet de lire la nouvelle valeur qui sera stockée dans la mémoire sélectionnée. Retapez la même valeur si vous ne voulez pas la modifier ;

- La routine ASCHEX convertit (ASCII->HEXA) les caractères ASCII entrés au clavier acquis dans la routine SAISIE. Le code HEXA résultant de la conversion est alors stocké dans la mémoire sélectionnée ;
- Les registres empilés en début de fonction sont dépilés et le contrôle est redonné au programme de menu.

## V. Affichage d'une zone mémoire : fonction DUMP

- Les registres AF, BC, DE et HL sont sauvegardés en début de routine ;
- L'adresse de début de DUMP est saisie grâce à la routine SP1 décrite précédemment ;
- L'affichage des mémoires commence à cette adresse et respecte le format suivant :

Adresse	Val	Val	Val	Val	Val	Val	Val	Val
Adresse	Val	Val	Val	Val	Val	Val	Val	Val
Adresse	Val	Val	Val	Val	Val	Val	Val	Val
Adresse	Val	Val	Val	Val	Val	Val	Val	Val
Adresse	Val	Val	Val	Val	Val	Val	Val	Val
Adresse	Val	Val	Val	Val	Val	Val	Val	Val
Adresse	Val	Val	Val	Val	Val	Val	Val	Val
Adresse	Val	Val	Val	Val	Val	Val	Val	Val
Adresse	Val	Val	Val	Val	Val	Val	Val	Val
Adresse	Val	Val	Val	Val	Val	Val	Val	Val

- Les registres empilés en début de fonction sont dépilés et le contrôle est redonné au programme de menu.

## VI. Retour au BASIC

Le programme debugger étant activé sous BASIC, il suffit d'une instruction RET pour y retourner. C'est ce que réalise l'option B du menu.

### Saisie du programme

Deux méthodes sont possibles :

- Saisie en assembleur ;
- Saisie en BASIC.

#### 1<sup>re</sup> méthode

Vous pouvez utiliser l'assembleur décrit dans cet ouvrage ou un autre assembleur pour entrer le programme « debugger ». Voici le listing commenté du programme en page 7.

```

1          ORG 8000H
2          LOAD 8000H
3          PRINT: EQU 0BB5AH          ;TXT OUTPUT
4          SUP:   EQU 62              ;Signe superieur
5          READ:  EQU 0BB06H         ;KM WAIT CHAR
6          CURS:  EQU 95              ;Caract. curseur
7          BS:    EQU 8               ;Caractere Back-Space
8          CR:    EQU 13              ;Carriage Return
9          LF:    EQU 10              ;Line Feed
10         BLANC: EQU 32              ;Caract. ESPACE
11         DEL:   EQU 127             ;Caract. DELETE
12         OPEN:  EQU 0BC77H         ;CAS IN OPEN
13         DREAD: EQU 0BC83H         ;CAS IN DIRECT
14         CLOSE: EQU 0BC7AH         ;CAS IN CLOSE
15         LECT:  EQU 82E0H          ;Routine externe
16         ECR:   EQU 835DH          ;Routine externe
17         ;=====
18         ;PROGRAMME PRINCIPAL
19         ;=====
20         BP:    EQU $               ;Boucle principale
21 8000 CDD480    CALL MENU            ;Affichage menu
22 8003 CDEB80    CALL CMD             ;Saisie d'une commande
23 8006 FE4C      CP 76                ;L?
24 8008 CAE082    JP Z,LECT            ;Lecture
25 800B FE45      CP 69                ;E?
26 800D CA5D83    JP Z,ECR             ;Ecriture
27 8010 FE58      CP 88                ;X?
28 8012 CA0F81    JP Z,EXEC            ;Execution
29 8015 FE4D      CP 77                ;M?
30 8017 CABF81    JP Z,MEM             ;Memoire
31 801A FE44      CP 68                ;D?
32 801C CA7081    JP Z,DUMP            ;Dump.

```

```
33 801F FE42          CP   66          ;B?
34 8021 CA4982       JP   Z,BASIC      ;Retour au BASIC
35 8024 18DA        JR   BP          ;Boucle principale
36 8026 00          NOP
37
38          BUFF:      EQU   $
39          ;-----
40          ;Definition de buffers
41          ;-----
42          DS   12          ;12 caract au maximum
43          MAX:      DS   1
44          ADDEB:    DS   2          ;Adresse de debut
45          ADFIN:    DS   2          ;Adresse de fin
46          BSTOP:    DS   3          ;Buffer stop prog.
47
48          DATA:    EQU   $
49          ;-----
50          ;Donnees alphanum. a afficher
51          ;-----
52 803B 44656275      DB   68,101,98,117,103,
52 803F 67676572
53 8043 205A3830      DB   32,90,56,48,13,10,
53 8047 0D0A4C29
53 804B 6563
54 804D 74757265      DB   116,117,114,101,32
54 8051 20202020
54 8055 20
55 8056 20452963      DB   32,69,41,99,114,10
55 805A 72697475
55 805E 72
56 805F 650D0A65      DB   101,13,10,101,88,4
56 8063 58296578
56 8067 63
```



```
57 8068 7574696F          DB  117,116,105,111,11
57 806C 6E202020
58 8070 4D29656D          DB  77,41,101,109,111,
58 8074 6F697265
59 8078 OD0A4429          DB  13,10,68,41,117,10
59 807C 756D7020
59 8080 20
60 8081 20202020          DB  32,32,32,32,32,32,
60 8085 20202042
60 8089 29
61 808A 61736963          DB  97,115,105,99,13,1
61 80BE OD0AFF
62
63
64          DATA2:      EQU  $              ;Message "Adresse :_"
65 8091 41647265          DB  65,100,114,101,115
65 8095 73736520
66 8099 3A5FFF           DB  58,95,OFFH
67
68          DATA3:      EQU  $              ;Message "a partir
69          ;de l'"
70 809C 61207061          DB  97,32,112,97,114,1
70 80A0 72746972
71 80A4 20646520          DB  32,100,101,32,108,
71 80AB 6C27FF
72
73          DATA4:      EQU  $              ;Message "jusqu'a l'"
74 80AB OD0A              DB  13,10
75 80AD 6A757371          DB  106,117,115,113,11
75 80B1 75276120
76 80B5 6C27FF           DB  108,39,OFFH
77
```

```

78          DATA5:      EQU  $                ;Message "Nom :_"
79 80B8 0D0A4E6F          DB  13,10,7B,111,109
79 80BC 6D
80 80BD 203A5FFF          DB  32,5B,95,0FFH
81
82          DATA6:      EQU  $                ;Message "A implanter
83                      ;a 1'"
84 80C1 0D0A4120          DB  13,10,65,32,105,10
84 80C5 696D
85 80C7 706C616E          DB  112,10B,97,110,116
85 80CB 74
86 80CC 65722061          DB  101,114,32,97,32,1
86 80D0 206C
87 80D2 27FF             DB  39,0FFH
88
89          ;-----
90          MENU:        EQU  $
91          ;-----
92          ;Entree: Donnees situees en DATA
93          ;Sortie: Aucun registre modifie
94          ;-----
95 80D4 213B80             LD  HL,DATA
96 80D7 0DDB80             CALL AFALPH          ;Affichage menu
97 80DA 09                RET
98
99
100         AFALPH:      EQU  $
101         ;-----
102         ;Entree : Zone d'aff dans HL
103         ;Sortie : Aucun registre modifie
104         ;-----

```

```

105 80DB E5          PUSH HL
106 80DC F5          PUSH AF
107                ME1:   EQU $                ;Boucle d'affichage
108 80DD 7E          LD   A,(HL)
109 80DE FEFF        CP   OFFH
110 80E0 2B06        JR   Z,ME2                ;Fin d'affichage
111 80E2 CD5ABB      CALL PRINT                ;Aff. caractere
112 80E5 23          INC  HL                ;Donnee suivante
113 80E6 1BF5        JR   ME1
114                ME2:   EQU $
115 80E8 F1          POP  AF
116 80E9 E1          POP  HL
117 80EA C9          RET
118
119                CMD:   EQU $
120                ;-----
121                ;Saisie d'une commande
122                ;Entree: Aucune
123                ;Sortie: Tous registres intacts
124                ;-----
125 80EB 3E3E        LD   A,SUP
126 80ED CD5ABB      CALL PRINT                ;Affichage ">"
127 80F0 3E5F        LD   A,CURS
128 80F2 CD5ABB      CALL PRINT                ;Affichage curseur
129 80F5 CD06BB      CALL READ                 ;Lecture d'un caract.
130 80F8 F5          PUSH AF
131 80F9 F5          PUSH AF
132 80FA 3E08        LD   A,BS
133 80FC CD5ABB      CALL PRINT                ;Affichage back-space
134 80FF F1          POP  AF
135 8100 CD5ABB      CALL PRINT                ;Affichage caractere
136 8103 3E0D        LD   A,CR

```

```

137 8105 CD5ABB          CALL PRINT
138 8108 3E0A          LD  A,LF
139 810A CD5ABB          CALL PRINT          ;Affiche CR+LF
140 810D F1            POP  AF
141 810E C9            RET
142
143          EXEC:      EQU  $
144          ;-----
145          ;Execution d'un programme
146          ;-----
147 810F F5            PUSH AF
148 8110 C5            PUSH BC
149 8111 D5            PUSH DE
150 8112 E5            PUSH HL          ;Sauvegarde des registres
151 8113 219C80        LD  HL,DATA3
152 8116 CDD880        CALL AFALPH
153 8119 CD9782        CALL SP1          ;Saisie de l'à de depart
154 811C 223480        LD  (ADDEB),HL   ;Sauv. à debut
155 811F 21A880        LD  HL,DATA4
156 8122 CDD880        CALL AFALPH
157 8125 CD9782        CALL SP1          ;Saisie de l'à de fin
158 8128 223680        LD  (ADFIN),HL   ;Sauvegarde à fin
159
160          X3:      EQU  $          ;Execution
161 812B 3E0D          LD  A,CR
162 812D CD5ABB          CALL PRINT
163 8130 3E0A          LD  A,LF
164 8132 CD5ABB          CALL PRINT          ;Passage a la ligne
165 8135 2A3680        LD  HL,(ADFIN)
166 8138 7E            LD  A,(HL)
167 8139 323880        LD  (BSTOP),A
168 813C 23            INC  HL

```

```

169 813D 7E          LD  A, (HL)
170 813E 323980     LD  (BSTOP+1), A
171 8141 23         INC  HL
172 8142 7E          LD  A, (HL)
173 8143 323A80     LD  (BSTOP+2), A      ;Sauvegarde code
174 8146 2A3680     LD  HL, (ADFIN)
175 8149 3EC3       LD  A, 0C3H
176 814B 77         LD  (HL), A
177 814C 3ED0       LD  A, 0D0H
178 814E 23         INC  HL
179 814F 77         LD  (HL), A
180 8150 3E83       LD  A, 83H           ;JP 83D0H
181 8152 23         INC  HL
182 8153 77         LD  (HL), A         ;Debranch en 9000H
183 8154 2A3480     LD  HL, (ADDEB)
184 8157 E9         JP  (HL)           ;Execution programme
185
186                ARET: EQU  $           ;Adresse de retour
187 8158 2A3680     LD  HL, (ADFIN)
188 815B 3A3880     LD  A, (BSTOP)
189 815E 77         LD  (HL), A
190 815F 3A3980     LD  A, (BSTOP+1)
191 8162 23         INC  HL
192 8163 77         LD  (HL), A
193 8164 3A3A80     LD  A, (BSTOP+2)
194 8167 23         INC  HL
195 8168 77         LD  (HL), A         ;Restitution code
196 8169 E1         POP  HL
197 816A D1         POP  DE
198 816B C1         POP  BC
199 816C F1         POP  AF           ;Restitution des registre
200 816D C30080     JP  BP           ;Retour boucle principale
201

```

```

202          DUMP:      EQU  $
203          ;-----
204          ;Affichage d'une zone memoire
205          ;-----
206 8170 F5          PUSH AF
207 8171 05          PUSH BC
208 8172 D5          PUSH DE
209 8173 E5          PUSH HL          ;Sauvegarde des registres
210 8174 CD9782     CALL SP1          ;Saisie de l'adresse
211 8177 1600      LD  D,0
212 8179 3E0D      DU1:  LD  A,CR
213 817B CD5ABB     CALL PRINT
214 817E 3E0A      LD  A,LF
215 8180 CD5ABB     CALL PRINT
216 8183 2A2980    LD  HL,(BUFF+2)    ;à le memoire
217 8186 7C        LD  A,H
218 8187 CD6682    CALL HEAS
219 818A 7D        LD  A,L
220 818B CD6682    CALL HEAS          ;Affichage adresse
221 818E 010000    LD  BC,0          ;Compteur colonnes
222 8191 3E20      DU2:  LD  A,BLANC
223 8193 CD5ABB     CALL PRINT          ;Affichage ESPACE
224 8196 2A2980    LD  HL,(BUFF+2)
225 8199 09        ADD HL,BC          ;à (mem) a afficher
226 819A 7E        LD  A,(HL)
227 819B CD6682    CALL HEAS          ;Aff. memoire
228 819E 03        INC  BC          ;Memoire suivante
229 819F 79        LD  A,C
230 81A0 FE08      CP  B
231 81A2 20ED      JR  NZ,DU2          ;Boucle sur colonne
232 81A4 23        INC  HL
233 81A5 222980    LD  (BUFF+2),HL    ;Sauv prochaine à

```

```

234 81A8 14          INC  D
235 81A9 7A          LD   A,D
236 81AA FE0A        CP   10
237 81AC 20CB        JR   NZ,DU1          ;Boucle sur ligne
238 81AE 3E0D        LD   A,CR
239 81B0 CD5ABB      CALL PRINT
240 81B3 3E0A        LD   A,LF
241 81B5 CD5ABB      CALL PRINT          ;Passage a la ligne
242 81B8 E1          POP  HL
243 81B9 D1          POP  DE
244 81BA C1          POP  BC
245 81BB F1          POP  AF          ;Restitution des registre
246 81BC C30080      JP   BP          ;Retour boucle principale
247
248                MEM:      EQU  $
249                ;-----
250                ;Lecture et/ou modification d'une
251                ;memoire
252                ;-----
253 81BF F5          PUSH AF
254 81C0 C5          PUSH BC
255 81C1 D5          PUSH DE
256 81C2 E5          PUSH HL          ;Sauvegarde des registres "
257 81C3 CD9782      CALL SP1          ;Saisie de l'adresse
258 81C6 7E          LD   A,(HL)
259 81C7 CD6682      CALL HEAS        ;Affichage (memoire)
260 81CA 3E20        LD   A,BLANC
261 81CC CD5ABB      CALL PRINT
262 81CF 3E5F        LD   A,CURS
263 81D1 CD5ABB      CALL PRINT
264
265 81D4 3E02        LD   A,2

```

```

266 81D6 323380      LD   (MAX),A          ;2 caracteres a saisir
267 81D9 CD0182      CALL SAISIE
268 81DC 212780      LD   HL,BUFF
269 81DF CD4A82      CALL ASCHEX          ;Conv. ASCII->HEXA
270 81E2 2A2980      LD   HL,(BUFF+2)
271 81E5 77          LD   (HL),A          ;Modif memoire
272 81E6 3E08        LD   A,BS
273 81E8 CD5ABB      CALL PRINT
274 81EB 3E20        LD   A,BLANC
275 81ED CD5ABB      CALL PRINT
276 81F0 3E0D        LD   A,CR
277 81F2 CD5ABB      CALL PRINT
278 81F5 3E0A        LD   A,LF
279 81F7 CD5ABB      CALL PRINT          ;Passage a la ligne
280 81FA E1          POP  HL
281 81FB D1          POP  DE
282 81FC C1          POP  BC
283 81FD F1          POP  AF          ;Retitution des registres
284 81FE C30080      JP   BP          ;Retour boucle principale
285
286
287                SAISIE:   EQU  $
288                ;-----
289                ;Saisie de caracteres alphanum.
290                ;Entree : (MAX)=nbre de caract
291                ;Sorcie : aucun registre ecrase
292                ;-----
293 8201 3A3380      LD   A,(MAX)
294 8204 57          LD   D,A          ;D=Nbre de caract a lire
295 8205 010000      LD   BC,0        ;Index dans le buffer
296 8208 212780      S1:   LD   HL,BUFF ;Buffer de lecture
297 820B CD06BB      CALL READ        ;1 caractere

```



```
298 820E FE7F          CP   DEL           ;DElete ?
299 8210 2817          JR   Z,S2          ;Traitement special
300 8212 F5            PUSH AF
301 8213 3E08          LD   A,BS
302 8215 CD5ABB        CALL PRINT
303 8218 F1            POP  AF
304 8219 CD5ABB        CALL PRINT
305 821C 09            ADD  HL,BC
306 821D 77            LD   (HL),A        ;Sauvegarde
307 821E 03            INC  BC
308 821F 3E5F          LD   A,CURS
309 8221 CD5ABB        CALL PRINT
310 8224 79            LD   A,C
311 8225 BA            CP   D
312 8226 20E0          JR   NZ,S1         ;Suite de la saisie
313 8228 C9            RET
314 8229 79            S2: LD   A,C
315 822A B7            OR   A
316 822B 28DB          JR   Z,S1           ;DEL non accepte
317 822D 0B            DEC  BC
318 822E 3E08          LD   A,BS
319 8230 CD5ABB        CALL PRINT         ;Retour en arriere
320 8233 3E20          LD   A,BLANC
321 8235 CD5ABB        CALL PRINT         ;Effacement caractere
322 8238 3E08          LD   A,BS
323 823A CD5ABB        CALL PRINT
324 823D 3E08          LD   A,BS
325 823F CD5ABB        CALL PRINT         ;2x retour en arriere
326 8242 3E5F          LD   A,CURS
327 8244 CD5ABB        CALL PRINT         ;Affichage curseur
328 8247 18BF          JR   S1
329
```

```

330          BASIC:      EQU  $
331 8249 C9          RET
332          ASCHEX:    EQU  $
333          ;-----
334          ;Conversion ASCII->HEXA
335          ;Entree : HL pointe sur le data
336          ;Sortie : A contient la conversion
337          ;          Autres registres intacts
338          ;-----
339 824A C5          PUSH BC          ;Sauvegarde
340 824B 7E          LD   A, (HL)
341 824C FE40        CP   40H
342 824E 3802        JR   C,AS1
343 8250 D607        SUB  7
344 8252 D630        AS1:  SUB  30H          ;Conversion MSQ
345 8254 17          RLA
346 8255 17          RLA
347 8256 17          RLA
348 8257 17          RLA          ;et passage en poids fort
349 8258 47          LD   B,A          ;Sauvegarde
350 8259 23          INC  HL
351 825A 7E          LD   A, (HL)
352 825B FE40        CP   40H
353 825D 3802        JR   C,AS2
354 825F D607        SUB  7
355 8261 D630        AS2:  SUB  30H          ;Conversion MSQ
356 8263 B0          OR   B          ;Octet complet
357 8264 C1          POP  BC          ;Restitution de BC
358 8265 C9          RET
359
360
361          HEAS:      EQU  $

```

```

362          ;-----
363          ;Routine de conversion HEXA->ASCII
364          ;et d'affichage du resultat
365          ;-----
366          ;Entree: A=nombre a convertir
367          ;Sortie: A efface
368          ;-----
369 8266 F5          PUSH AF          ;Sauvegarde de A
370 8267 E6F0       AND  OF0H        ;Isole MSQ
371 8269 1F         RRA
372 826A 1F         RRA
373 826B 1F         RRA
374 826C 1F         RRA          ;Isole MSQ
375 826D FE0A       CP   10          ;Chiffre?
376 826F 3802       JR   C,HEAS1     ;Oui
377 8271 C607       ADD  A,7          ;Decalage pour une lettre
378          HEAS1: EQU  $
379 8273 C630       ADD  A,30H        ;Conversion
380 8275 CD5ABB     CALL PRINT       ;Affichage MSQ
381 8278 F1         POP  AF          ;A retrouve
382 8279 E60F       AND  OFH
383 827B CB17       RL   A
384 827D CB17       RL   A
385 827F CB17       RL   A
386 8281 CB17       RL   A          ;Isole LSQ
387 8283 CB07       RLC  A
388 8285 CB07       RLC  A
389 8287 CB07       RLC  A
390 8289 CB07       RLC  A          ;LSQ a droite
391 828B FE0A       CP   10          ;Chiffre?
392 828D 3802       JR   C,HEAS2     ;Oui
393 828F C607       ADD  A,7

```

```
394          HEAS2:      EQU  $
395 8291 C630          ADD  A,30H          ;Conversion
396 8293 CD5ABB          CALL PRINT          ;Affichage LSQ
397 8296 C9            RET
398
399
400          ;-----
401          ; ZONE SOUS PROGRAMMES
402          ;-----
403          SP1:        EQU  $          ;Saisie d'une adresse
404 8297 219180         LD   HL,DATA2
405 829A CDD880         CALL AFALPH          ;Affichage message
406 829D 3E04          LD   A,4
407 829F 323380         LD   (MAX),A          ;4 caract a saisir
408 82A2 CD0182         CALL SAISIE          ;Saisie
409 82A5 3E08          LD   A,BS
410 82A7 CD5ABB          CALL PRINT
411 82AA 3E20          LD   A,BLANC
412 82AC CD5ABB          CALL PRINT
413 82AF 212780         LD   HL,BUFF
414 82B2 CD4A82         CALL ASCHEX          ;Conversion ASCII->HEX
415 82B5 322780         LD   (BUFF),A          ;Sauvegarde LSB
416 82B8 212980         LD   HL,BUFF+2
417 82BB CD4A82         CALL ASCHEX
418 82BE 6F            LD   L,A
419 82BF 3A2780         LD   A,(BUFF)
420 82C2 67            LD   H,A          ;HL=à memoire
421 82C3 222980         LD   (BUFF+2),HL      ;Sauvegarde à
422 82C6 C9            RET
423          END
```

```
1          ORG  82E0H
2          LOAD 82E0H
3
4          BUFF: EQU  8027H
5          OPEN: EQU  0BC77H      ;Duv. fichier
6          DREAD: EQU  0BC83H    ;Lecture disque
7          CLOSE: EQU  0BC7AH    ;Ferm. fichier
8          DATA6: EQU  80C1H    ;Message externe
9          AFALPH: EQU  80DBH    ;Routine externe
10         SP1:   EQU  8297H     ;Routine externe
11         CR:   EQU  13        ;Carriage Return
12         LF:   EQU  10        ;Line Feed
13         PRINT: EQU  0BB5AH   ;TXT OUTPUT
14         BP:   EQU  8000H     ;Boucle principale
15         DATA5: EQU  80B8H   ;Message externe
16         READ: EQU  0BB06H   ;KM WAIT CHAR
17         DEL:  EQU  127      ;DELeTe
18         BS:   EQU  8        ;Back Space
19         CURS: EQU  95       ;Caractere curseur
20         BLANC: EQU  32      ;Caractere espace
21         WOPEN: EQU  0BC8CH   ;CAS OUT OPEN
22         WRITE: EQU  0BC98H   ;CAS OUT DIRECT
23         WCLOSE: EQU  0BC8FH  ;CAS OUT CLOSE
24         DATA3: EQU  809CH   ;Message externe
25         ADDEB: EQU  8034H    ;à debut exec.
26         ADFIN: EQU  8036H    ;à fin exec.
27         DATA4: EQU  80ABH   ;Message externe
28
29         LECT: EQU  $
30         ;-----
31         ;Lecture d'un programme objet
32         ;-----
```

```

33 82E0 F5          PUSH AF
34 82E1 C5          PUSH BC
35 82E2 D5          PUSH DE
36 82E3 E5          PUSH HL          ;Sauvegarde des registres
37 82E4 CD1183      CALL SAINOM      ;Saisie du nom
38 82E7 41          LD B,C          ;Longueur du nom
39 82E8 1100C0      LD DE,0C000H    ;Buffer de 2 KO
40 82EB 212780      LD HL,BUFF      ;Zone du nom
41 82EE CD77BC      CALL OPEN       ;Ouverture fichier
42 82F1 21C180      LD HL,DATAG
43 82F4 CDD880      CALL AFALPH     ;à d'i:plantation
44 82F7 CD9782      CALL SP1
45 82FA 3E0D        LD A,CR
46 82FC CD5ABB      CALL PRINT
47 82FF 3E0A        LD A,LF
48 8301 CD5ABB      CALL PRINT      ;CR+LF
49                ;HL contient l'à d'implantation
50 8304 CDB3BC      CALL DREAD      ;Lecture disque
51 8307 CD7ABC      CALL CLOSE      ;Fermeture fichier
52 830A E1          POP HL
53 830B D1          POP DE
54 830C C1          POP BC
55 830D F1          POP AF          ;Restitution des registre
56 830E C30080      JP BP           ;Retour boucle principale
57
58                SAINOM: EQU $          ;Saisie du nom
59                ;-----
60                ;Saisie de caracteres alphanum.
61                ;se terminant par un CR dans une
62                ;limite de 12 caracteres
63                ;-----
64 8311 21B880      LD HL,DATAS

```

```
65 8314 CDDB80          CALL AFALPH          ;Saisie du nom
66
67 8317 010000         LD   BC,0
68 831A 2127B0   L1:   LD   HL,BUFF          ;Buffer de lecture
69 831D CD06BB         CALL READ
70 8320 FE7F         CP   DEL
71 8322 2818         JR   Z,L2          ;Traitement special
72 8324 FE0D         CP   CR
73 8326 2834         JR   Z,L3          ;Fin de saisie
74 8328 F5          PUSH AF
75 8329 3E08         LD   A,BS
76 832B CD5ABB         CALL PRINT
77 832E F1          POP  AF
78 832F CD5ABB         CALL PRINT
79 8332 09          ADD  HL,BC
80 8333 77          LD   (HL),A          ;Sauvegarde
81 8334 03          INC  BC
82 8335 3E5F         LD   A,CURS
83 8337 CD5ABB         CALL PRINT
84 833A 18DE         JR   L1
85 833C 79          L2:   LD   A,C
86 833D B7          OR   A
87 833E 28DA         JR   Z,L1
88 8340 0B          DEC  BC
89 8341 3E08         LD   A,BS
90 8343 CD5ABB         CALL PRINT          ;Retour en arriere
91 8346 3E20         LD   A,BLANC
92 8348 CD5ABB         CALL PRINT          ;Effacement caractere
93 834B 3E08         LD   A,BS
94 834D CD5ABB         CALL PRINT
95 8350 3E08         LD   A,BS
96 8352 CD5ABB         CALL PRINT          ;2x retour en arriere
```

```

97 8355 3E5F          LD  A,CURS
98 8357 CD5ABB        CALL PRINT           ;Affichage curseur
99 835A 18BE          JR   L1
100 835C C9           L3:   RET           ;Fin de saisie
101
102
103                   ECR:   EQU  $
104                   ;-----
105                   ;Sauvegarde d'1 programme objet
106                   ;-----
107 835D F5           PUSH AF
108 835E C5           PUSH BC
109 835F D5           PUSH DE
110 8360 E5           PUSH HL           ;Sauvegarde des registres
111 8361 CD1183        CALL SAINOM         ;Saisie du nom
112 8364 41           LD   B,C           ;Longueur nom
113 8365 1100C0        LD   DE,0C000H     ;Buffer 2 KO
114 8368 212780        LD   HL,BUFF       ;Zone du nom
115 836B CD8CBC        CALL WOPEN
116
117 836E 3E0D          LD   A,CR
118 8370 CD5ABB        CALL PRINT
119 8373 3E0A          LD   A,LF
120 8375 CD5ABB        CALL PRINT         ;CR+LF
121 8378 219C80        LD   HL,DATA3
122 837B CDD880        CALL AFALPH
123 837E CD9782        CALL SP1
124 8381 223480        LD   (ADDEB),HL   ;Sauv à debut
125 8384 21AB80        LD   HL,DATA4
126 8387 CDD880        CALL AFALPH
127 838A CD9782        CALL SP1
128 838D 223680        LD   (ADFIN),HL   ;Sauv à fin

```



```
129 8390 2A3680          LD   HL, (ADFIN)
130 8393 ED5B3480       LD   DE, (ADDEB)
131 8397 37             SCF
132 8398 3F             CCF                ;Carry a zero
133 8399 ED52          SBC  HL, DE
134 839B 223680        LD   (ADFIN),HL    ;Sauv longueur
135
136 839E 2A3480        LD   HL, (ADDEB)
137 83A1 ED5B3680       LD   DE, (ADFIN)
138 83A5 ED4B3480       LD   BC, (ADDEB)
139 83A9 3E2A          LD   A,2AH        ;Fichier binaire
140 83AB CD98BC         CALL WRITE
141 83AE CDBFBC         CALL WCLOSE
142 83B1 3E0D          LD   A,CR
143 83B3 CD5ABB         CALL PRINT
144 83B6 3E0A          LD   A,LF
145 83B8 CD5ABB         CALL PRINT        ;Passage a la ligne
146 83BB E1            POP  HL
147 83BC D1            POP  DE
148 83BD C1            POP  BC
149 83BE F1            POP  AF          ;Restitution des registre
150 83BF C30080        JP   BP          ;Retour boucle principale
151                    END
```

```

1          ORG 83D0H
2          LOAD 83D0H
3          ;
4          BLANC: EQU 32          ;Code HEXA espace
5          PRINT: EQU 0BB5AH     ;TXT OUTPUT
6          CR: EQU 13           ;Carriage Return
7          LF: EQU 10           ;Line Feed
8          ARET: EQU 8158H      ;Adresse de retour
9          ADFIN: EQU 8036H     ;à fin execution
10
11         AREG: EQU $
12         ;=====
13         ;AFFICHAGE DES REGISTRES
14         ;=====
15 83D0 CD1184          CALL ALPHA          ;Affichage alphanum.
16 83D3 CD2484          CALL AVAL           ;Affichage des valeurs
17 83D6 C35881          JP ARET            ;Retour au prog principal
18 83D9 C9             RET
19         ;=====
20
21
22
23
24         DATA: EQU $
25         ;-----
26         ;Donnees alphanum. a afficher
27         ;-----
28 83DA 0D0A           DB 13,
29 83DC 20484C20       DB 20H,48H,4CH,20H
30 83E0 20204445       DB 20H,20H,44H,45H
31 83E4 20202042       DB 20H,20H,20H,42H
32 83E8 43202020       DB 43H,20H,20H,20H

```

```

33 83EC 41462020          DB   41H,46H,20H,20H
34 83F0 2052490A          DB   20H,52H,49H,0AH
35 83F4 0A0A0D20          DB   0AH,0AH,0DH,20H
36 83F8 49582020          DB   49H,58H,20H,20H
37 83FC 20495920          DB   20H,49H,59H,20H
38 8400 20205350          DB   20H,20H,53H,50H
39 8404 20202050          DB   20H,20H,20H,50H
40 8408 430B0B0D          DB   43H,0BH,0BH,0DH
41 840C FF                DB   OFFH

42          ;-----
43          ;Zone buffer
44          ;-----
45          BUFF:        DS   2
46          SPC:         DS   2
47          ;
48          ALPHA:      EQU  $
49          ;-----
50          ;Affichage de texte
51          ;Entree : donnees situees en DATA
52          ;Sortie : aucun registre efface
53          ;-----
54 8411 E5                PUSH HL
55 8412 F5                PUSH AF          ;Sauvegardes
56 8413 21DAB3           LD    HL,DATA    ;HL pointe su 1e data
57          AF1:         EQU  $          ;Boucle d'affichage
58 8416 7E                LD    A,(HL)
59 8417 FEFF             CP    OFFH
60 8419 2806             JR    Z,AF2    ;Fin d'affichage
61 841B CD5ABB           CALL PRINT      ;Affichage
62 841E 23                INC   HL
63 841F 18F5             JR    AF1
64          AF2:         EQU  $

```

```
65 8421 F1          POP  AF
66 8422 E1          POP  HL          ;Restitution registres
67 8423 C9          RET
68
69
70          AVAL:      EQU  $
71          ;-----
72          ;Aff. des valeurs des registres
73          ;Entree : Aucune
74          ;Sortie : Aucun registre ecrase
75          ;-----
76 8424 E5          PUSH HL
77 8425 F5          PUSH AF
78 8426 F5          PUSH AF          ;Sauvegarde registres
79 8427 7C          LD   A,H
80 8428 CD1385      CALL HEAS
81 842B 7D          LD   A,L
82 842C CD1385      CALL HEAS          ;Affichage HL
83 842F 3E20        LD   A,BLANC
84 8431 CD5ABB      CALL PRINT          ;Affichage espace
85 8434 7A          LD   A,D
86 8435 CD1385      CALL HEAS
87 8438 7B          LD   A,E
88 8439 CD1385      CALL HEAS          ;Affichage DE
89 843C 3E20        LD   A,BLANC
90 843E CD5ABB      CALL PRINT          ;Affichage espace
91 8441 7B          LD   A,B
92 8442 CD1385      CALL HEAS
93 8445 79          LD   A,C
94 8446 CD1385      CALL HEAS          ;Affichage BC
95 8449 3E20        LD   A,BLANC
96 844B CD5ABB      CALL PRINT          ;Affichage espace
```

```
97 844E E1      POP  HL
98 844F 7C      LD   A,H
99 8450 CD1385  CALL HEAS
100 8453 7D     LD   A,L
101 8454 CD1385  CALL HEAS          ;Affichage AF
102 8457 3E20   LD   A,BLANC
103 8459 CD5ABB  CALL PRINT        ;Affichage espace
104 845C ED5F   LD   A,R
105 845E CD1385  CALL HEAS
106 8461 ED57   LD   A,I
107 8463 CD1385  CALL HEAS          ;Affichage RI
108
109 8466 3E0D   LD   A,CR
110 8468 CD5ABB  CALL PRINT
111 846B 3E0A   LD   A,LF
112 846D CD5ABB  CALL PRINT        ;CR+LF
113 8470 D9     EXX
114 8471 7C     LD   A,H
115 8472 D9     EXX
116 8473 CD1385  CALL HEAS
117 8476 D9     EXX
118 8477 7D     LD   A,L
119 8478 D9     EXX
120 8479 CD1385  CALL HEAS          ;Affichage HL'
121 847C 3E20   LD   A,BLANC
122 847E CD5ABB  CALL PRINT        ;Affichage espace
123 8481 D9     EXX
124 8482 7A     LD   A,D
125 8483 D9     EXX
126 8484 CD1385  CALL HEAS
127 8487 D9     EXX
128 8488 7B     LD   A,E
```

```
129 8489 D9          EXX
130 848A CD1385      CALL HEAS           ;Affichage DE'
131 848D 3E20        LD  A,BLANC
132 848F CD5ABB      CALL PRINT         ;Affichage espace
133 8492 D9          EXX
134 8493 78          LD  A,B
135 8494 D9          EXX
136 8495 CD1385      CALL HEAS
137 8498 D9          EXX
138 8499 79          LD  A,C
139 849A D9          EXX
140 849B CD1385      CALL HEAS           ;Affichage BC'
141 849E 3E20        LD  A,BLANC
142 84A0 CD5ABB      CALL PRINT         ;Affichage espace
143 84A3 08          DB  08             ;EX AF,AF'
144 84A4 E5          PUSH HL
145 84A5 F5          PUSH AF
146 84A6 CD1385      CALL HEAS           ;Affichage A'
147 84A9 E1          POP  HL
148 84AA 7D          LD  A,L
149 84AB CD1385      CALL HEAS           ;Affichage F'
150 84AE 08          DB  08             ;EX AF,AF'
151 84AF E1          POP  HL
152 84B0 3E0D        LD  A,CR
153 84B2 CD5ABB      CALL PRINT
154 84B5 3E0A        LD  A,LF
155 84B7 CD5ABB      CALL PRINT
156 84BA 3E0A        LD  A,LF
157 84BC CD5ABB      CALL PRINT         ;CR+LF+LF
158 84BF DD220D84    LD  (BUFF),IX
159 84C3 2A0D84      LD  HL,(BUFF)
160 84C6 7C          LD  A,H
```

```
161 84C7 CD1385      CALL HEAS
162 84CA 7D         LD  A,L
163 84CB CD1385      CALL HEAS           ;Affichage IX
164 84CE 3E20       LD  A,BLANC
165 84D0 CD5ABB      CALL PRINT          ;Affichage espace
166 84D3 FD220D84    LD  (BUFF),IY
167 84D7 2A0D84     LD  HL,(BUFF)
168 84DA 7C         LD  A,H
169 84DB CD1385      CALL HEAS
170 84DE 7D         LD  A,L
171 84DF CD1385      CALL HEAS           ;Affichage IY
172 84E2 3E20       LD  A,BLANC
173 84E4 CD5ABB      CALL PRINT          ;Affichage espace
174 84E7 ED730D84    LD  (BUFF),SP
175 84EB 2A0D84     LD  HL,(BUFF)
176 84EE 7C         LD  A,H
177 84EF CD1385      CALL HEAS
178 84F2 7D         LD  A,L
179 84F3 CD1385      CALL HEAS           ;Affichage SP
180 84F6 3E20       LD  A,BLANC
181 84F8 CD5ABB      CALL PRINT          ;Affichage espace
182 84FB 2A3680     LD  HL,(ADFIN)
183 84FE 7C         LD  A,H
184 84FF CD1385      CALL HEAS
185 8502 7D         LD  A,L
186 8503 CD1385      CALL HEAS           ;Affichage PC
187 8506 3E0D       LD  A,CR
188 8508 CD5ABB      CALL PRINT
189 850B 3E0A       LD  A,LF
190 850D CD5ABB      CALL PRINT          ;CR+LF
191 8510 F1         POP  AF
192 8511 E1         POP  HL
```

```
193 8512 C9          RET
194
195
196
197          HEAS:      EQU  $
198          ;-----
199          ;Conversion HEXA -> ASCII et
200          ;affichage du resultat
201          ;-----
202          ;Entree: A=Nombre a convertir
203          ;Sortie: A efface
204          ;-----
205 8513 F5          PUSH AF          ;Sauvegarde de A
206 8514 E6F0        AND  OF0H          ;Isole MSQ
207 8516 1F          RRA
208 8517 1F          RRA
209 8518 1F          RRA
210 8519 1F          RRA          ;Isole MSQ
211 851A FE0A        CP   10          ;Chiffre ?
212 851C 3802        JR   C,HEAS1     ;Oui
213 851E C607        ADD  A,7          ;Decalage pour une lettre
214          HEAS1:    EQU  $
215 8520 C630        ADD  A,30H         ;Conversion
216 8522 CD5ABB      CALL PRINT        ;Affichage MSQ
217 8525 F1          POP  AF          ;A retrouve
218 8526 E60F        AND  OFH          ;Isole LSQ
219 8528 CB17        RL   A
220 852A CB17        RL   A
221 852C CB17        RL   A
222 852E CB17        RL   A          ;Isole,LSQ
223 8530 CB07        RLC  A
224 8532 CB07        RLC  A
```



```
225 8534 CB07          RLC  A
226 8536 CB07          RLC  A          ;LS0 a droite
227 8538 FE0A          CP   10          ;Chiffre ?
228 853A 3802          JR   C,HEAS2      ;Oui
229 853C C607          ADD  A,7
230                   HEAS2:  EQU  $
231 853E C630          ADD  A,30H        ;Conversion
232 8540 CD5ABB        CALL PRINT      ;Affichage
233 8543 C9           RET
234                   END
```

Une fois le programme saisi et assemblé, vous sauvegarderez la zone mémoire comprise entre #8000 et #8543. Le programme BASIC suivant permettra d'activer le debugger :

```
10 MEMORY &5000
20 LOAD"DEBUG.BIN"
30 CALL &8000
```

## 2<sup>e</sup> méthode

Si vous préférez saisir le programme debugger sous BASIC, il vous faudra procéder en deux étapes :

- 1) Saisie d'un programme qui contient les codes HEXA du programme debugger et qui fabrique le programme binaire DEBUG.BIN (Cf. programme page suivante).
- 2) Activation du debugger par le même programme BASIC que celui décrit page suivante.





